

Afortunadamente hay gente para todo, diversa, con gustos sorprendentes. Hay a quienes les gusta la programación. Escriben código en algún lenguaje y, como resultado, ocurren cosas. Pero no todo el mundo es así. **NIFO** es un lenguaje de programación de cuestionarios, y no solo cuestionarios 🤖. Eso da mucha potencia y libertad. Pero también excluye a quienes no gustan de escribir código porque tienen otras cosas más importantes a las que dedicarse. Así que hemos pensado en...



Lenguaje Derivado

[Para situarnos](#) (1) - [Formato de pantalla antigua](#) (3) - [Formato atrevido](#) (4)

Para comprender bien este documento es necesario que ya conozcas el lenguaje **NIFO**, que hayas hecho algo con él, algún cuestionario por ejemplo. Si no es así, te aconsejo leer primero el **Manual de referencia** o bien **Mi primer cuestionario NIFO**.

Para situarnos

[↑ Subir](#)

Algo con Likert

Imagina que deseas escribir un ítem que tiene un formato tipo Likert. Un ejemplo concreto para ese objetivo es este:

```
@Spaco 7;  
@Teksto ¿Te gusta la quinoa con lechuga? Indica a continuación cuánto te gusta, escogiendo el punto del continuo que mejor lo expresa;  
@Spaco 3;  
@demLikert {var1;Nada de nada;7;Me encanta!}
```

El resultado, si todas las demás opciones de **NIFO** son las definidas por defecto, es el siguiente:

¿Te gusta la quinoa con lechuga? Indica a continuación cuánto te gusta, escogiendo el punto del continuo que mejor lo expresa:

Nada de nada Me encanta!

Si hay que repetir el mismo sistema varias veces, merece la pena definir una función o variable compleja. Por ejemplo así:

```
@Difino puntos {7}
@Difino likert (variable,pregunta) {
@Spaco 7; @Teksto $pregunta; @Spaco 3;
@demLikert{$variable;Nada de nada;$puntos;Me encanta!}
}
```

Y podríamos escribir esto:

```
$likert {var1::¿Te gusta la quinoa con lechuga? Indica a continuación
cuánto te gusta, escogiendo el punto del continuo que mejor lo expresa:}
$likert {var2::¿Y las acelgas con champán?}
$likert {var3::Y, para terminar, ¿qué te parece unas anchoas con batido de
cacao?}
```

El resultado sería el siguiente:

¿Te gusta la quinoa con lechuga? Indica a continuación cuánto te gusta, escogiendo el punto del continuo que mejor lo expresa:

Nada de nada Me encanta!

¿Y las acelgas con champán?

Nada de nada Me encanta!

Y, para terminar, ¿qué te parece unas anchoas con batido de cacao?

Nada de nada Me encanta!

Hemos ahorrado mucho tecleo de código. Observa que todos los ítems con formato Likert tienen que ser de 7 puntos, que el espacio entre el texto y el formato es siempre de 3 unidades, y que entre un ítem y el siguiente median 7 puntos de espacio.

No obstante, la libertad no se ha perdido. Si quieres contruir muchos ítems con esas características, tienes la función `$likert`. Pero si quieres modificar algo, siempre podrás acudir al código **NIFO** directo.

Principio general de un lenguaje derivado

Con **NIFO** puedes conseguir una gran diversidad de efectos, con cuestionarios personalizados, que tienen tu estilo. Llegar a construir instrumentos de medida con un alto nivel de detalle puede exigir muchas líneas de código. Y es una elección personal.

Es posible que tengas ya un estilo propio y que te guste que tus cuestionarios lo muestren. Los puntos de opción serán cuadrados y gruesos; Likert con 5 puntos con las opciones extremas representadas por iconos de tristeza y felicidad; distancias de 8 unidades entre dos ítems; ítems marcados dentro de recuadros de borde azul; etc. Si es así, puedes definir un metalenguaje o un lenguaje derivado, es decir, una colección de definiciones de variables y funciones mediante el comando `@difino`, de tal forma que ahorres mucho trabajo de código después.

En lo que sigue vamos a ver algunos ejemplos. Son ejemplos sencillos para no saturar el documento con demasiadas líneas de código. A pesar de su sencillez, pueden sugerirte ideas, estimular tu imaginación para que puedas encontrar tu estilo personal u ofrecer a investigadores una forma peculiar, bella o distinta de dar formato a los cuestionarios.

Disfruta.

Vamos a diferenciar entre la especificación del estilo y el contenido en sí del cuestionario. Comencemos por el estilo:

```
/*
   Definición del estilo
   -----
*/

// Ventana
@mFenestro 20;
@kFenestro 0,0,0;
@kTeksto 255,255,255;
@tTeksto Cutive mono;

// Detalles
@difino l {@linio 100;}
@difino previo (texto) {
  @Spaco 10; $l @Spaco 2;
  @Teksto @t{k;$texto};
  @Spaco 2; $l}
@difino unue (a) {@Spaco 8; @Teksto ■ $a; @Spaco 2;}
@difino pregunta (variable,texto) {
  $unue {$texto}
  @demLikert {$variable;En desacuerdo;5;De acuerdo}}
@difino frecuencia (variable,texto) {
  $unue {$texto}
  @demListo {$variable;ne;Nunca;1 vez al mes;1 vez a la semana; Diariamente}}
```

A ello le añadimos un contenido específico. Por ejemplo:

```
// Contenido
$previo {Expresa tu grado de acuerdo con las siguientes afirmaciones}
$pregunta {uno::Me gusta el chocolate con 100% de cacao}
$pregunta {dos::Si pudiera, me gustaría hacer un viaje espacial}
$pregunta {tres::En términos generales, me gusta la vida que llevo}

$previo {A continuación indica con qué frecuencia haces cada una
  de las siguientes actividades. Escoge la frecuencia que más se
  aproxima a tu hábito o realidad.}
$frecuencia {cuatro::Usar la bicicleta}
$frecuencia {cinco::Ver series}
$frecuencia {seis::Visitar a la familia}
```

El resultado lo tienes en la siguiente página.

Observa que hemos definido primero una apariencia general del cuestionario: márgenes de 20 décimas de em ([@mFenestro 20](#)), fondo negro ([@kFenestro 0,0,0;](#)) y letra en blanco ([@kTeksto 255,255,255;](#)) con tipografía Cutive mono ([@tTeksto Cutive mono;](#)).

A continuación se han definido los detalles de formato:

- Las líneas van a ser siempre completas, es decir, del 100% ([@linio 100](#))
- El texto previo a los grupos de ítems ([\\$previo](#)) consiste en un espacio de 10 unidades em ([@Spaco 10](#)) y un texto en cursivas ([@Teksto @t{k;\\$texto}](#)), antecedido y seguido de una línea completa ([\\$l](#)) con una separación de 2 unidades ([@Spaco 2](#)).
- Los textos de los ítems van a seguir siempre el mismo esquema, gracias a la función [\\$unue](#): un espacio de 8 décimas de em ([@Spaco 8](#)), el texto antecedido por un recuadro ■ ([@Teksto ■ \\$a](#)), y un espacio final de 2 décimas ([@Spaco 2](#)).

- Se manejan dos tipos de formatos de ítem. Por un lado el tipo Likert, con escala de acuerdo de cinco puntos (`@demLikert {$variable;En desacuerdo;5;De acuerdo}`).
- Por otro, formato de listado de opciones de frecuencia incompatibles en línea (`@demListo {$variable;ne;Nunca;1 vez al mes;1 vez a la semana; Diariamente}`).

```

Expresa tu grado de acuerdo con las siguientes afirmaciones

■ Me gusta el chocolate con 100% de cacao
    En desacuerdo ○ ○ ○ ○ De acuerdo

■ Si pudiera, me gustaría hacer un viaje espacial
    En desacuerdo ○ ○ ○ ○ De acuerdo

■ En términos generales, me gusta la vida que llevo
    En desacuerdo ○ ○ ○ ○ De acuerdo

A continuación indica con qué frecuencia haces cada una de las
siguientes actividades. Escoge la frecuencia que más se aproxima
a tu hábito o realidad.

■ Usar la bicicleta
    ○ Nunca ○ 1 vez al mes ○ 1 vez a la semana ○ Diariamente

■ Ver series
    ○ Nunca ○ 1 vez al mes ○ 1 vez a la semana ○ Diariamente

■ Visitar a la familia
    ○ Nunca ○ 1 vez al mes ○ 1 vez a la semana ○ Diariamente

```

Formato atrevido

[↑Subir](#)

Para jugar con las posibilidades, imaginemos ahora un estilo más chillón, con colores llamativos de la gama ocre, recuadros para las preguntas y diferentes grosores del texto según su función.

No voy a pararme en los detalles del código. Si has entendido sin problemas la porción de código del ejemplo anterior, no tendrás problemas con este. En cualquier caso, sabes que puedes contar con el manual de referencia.

El contenido del cuestionario va a ser el mismo, por lo que no hay modificaciones que mencionar desde la etiqueta

```
// Contenido
```

Lo diferente se encuentra en las especificaciones de la ventana y de las funciones \$previo, \$pregunta y \$frecuencia, que son llamadas desde el código del contenido del cuestionario. Veamos primero el resultado y, después, el código específico.

He aquí el resultado:

Expresa tu grado de acuerdo con las siguientes afirmaciones

Me gusta el chocolate con 100% de cacao
En desacuerdo De acuerdo

Si pudiera, me gustaría hacer un viaje espacial
En desacuerdo De acuerdo

En términos generales, me gusta la vida que llevo
En desacuerdo De acuerdo

A continuación indica con qué frecuencia haces cada una de las siguientes actividades. Escoge la frecuencia que más se aproxima a tu hábito o realidad.

Usar la bicicleta
 Nunca 1 vez al mes 1 vez a la semana Diariamente

Ver series
 Nunca 1 vez al mes 1 vez a la semana Diariamente

Visitar a la familia
 Nunca 1 vez al mes 1 vez a la semana Diariamente

Y aquí el código de formato que lo ha generado:

```
// Ventana
@mFenestro 20;
@kFenestro 255,177,0;
@kTeksto 0,0,100;
@tTeksto Parisienne;

// Detalles
@difino e2 {@Spaco 2;}
@difino e5 {@Spaco 5;}
@difino previo (texto) {
  @Spaco 10; $e2
  @Teksto @t{krd;$texto}; $e5
@difino unue (a) {@Teksto @t{d;$a}; $e2}
@difino pregunta (variable,texto) {
  @Bloko {
    @mBloko 5; @rBordo 5;
    @kBloko 255,200,50;
```

```
$unue {$texto}
  @demLikert {$variable;En desacuerdo;5;De acuerdo}
} $e5
}
@difino frecuencia (variable,texto) {
  @Bloko {
    @mBloko 5; @rBordo 5;
    @kBloko 255,200,50;
    $unue {$texto}
    @demListo {$variable;ne;Nunca;1 vez al mes;1 vez a la semana; Diariamente}
  } $e5
}
```